

Глава 5. Оператор конкатенации массивов

Это глава из книги: Майкл Гирвин. Ctrl+Shift+Enter. Освоение формул массива в Excel.

[Предыдущая глава](#)

[Оглавление](#)

[Следующая глава](#)

Оператор конкатенации – амперсанд & – позволяет объединить два элемента так, чтобы они стали единым целым. Вы можете соединять числа, текст (в кавычках), результаты формулы и др. Вы увидите, как использовать амперсанд, чтобы создать оператор массива, содержащий поиск по двум критериям.

На рис. 5.1 показаны исходные данные в диапазоне A2:C16 и отчет, который вы хотите создать в диапазоне E2:G10. Кросс-табличный отчет должен показывать объем продаж для каждого кода товара и соответствующего типа: левостороннего (L) и правостороннего (R). Для расчетов вы хотите создать формулу в ячейке F4, а затем скопировать ее в диапазон F4:G10. Проблема в том, что для каждой ячейки в диапазоне F4:G10, у вас есть два критерия поиска. Например, в ячейке F4, объем 30 взят на основе значения кода товара в ячейке E4 и значения L в ячейке F3. Стандартные функции поиска Excel запрограммированы так, чтобы искать только одно значение в одном столбце. Один из способов решения задачи – объединить два критерия поиска в один внутри формулы. (Предполагается, что таблица A2:C16 содержит по одной комбинации кода и типа товара, например, 2A35-2A36 типа L представлен в таблице одной строкой.)

F4										
: {=ИНДЕКС(\$C\$3:\$C\$16;ПОИСКПОЗ(\$E4&F\$3;\$A\$3:\$A\$16&\$B\$3:\$B\$16;0))}										
	A	B	C	D	E	F	G	H	I	J
1	Создание таблицы, которая ищет объем продаж на основе двух критериев									
2	Код товара	L/R?	Кол-во			Кол-во				
3	2A45-2A46	R	18		Код товара	L	R			
4	2A39-2A40	L	36		2A35-2A36	30	35			
5	2A35-2A36	L	30		2A38-2A39	30	36			
6	2A38-2A39	R	36		2A39-2A40	36	35			
7	2A44-2A45	L	18		2A44-2A45	18	28			
8	2A48-2A49	L	20		2A45-2A46	24	18			
9	2A45-2A46	L	24		2A46-2A47	30	36			
10	2A48-2A49	R	24		2A48-2A49	20	24			
11	2A44-2A45	R	28							
12	2A46-2A47	L	30							
13	2A38-2A39	L	30							
14	2A35-2A36	R	35							
15	2A39-2A40	R	35							
16	2A46-2A47	R	36							
17										

Рис. 5.1. Цель – создать перекрестную таблицу, основанную на двух критериях поиска

Вы можете использовать функцию поиска ИНДЕКС. В качестве аргумента массив выберите диапазон значений, которые вы хотите извлечь; в нашем случае – это C3:C16 (рис. 5.2). Поскольку мы будем протягивать нашу формулу, мы подготовимся заранее и введем диапазон \$C\$3:\$C\$16 в абсолютных ссылках, чтобы он не исказился при протягивании формулы.

Код товара	L	R
2A35-2A36	=ИНДЕКС(\$C\$3:\$C\$16	
2A38-2A39	ИНДЕКС(массив; номер_строки; [номер_столбца])	
2A39-2A40	ИНДЕКС(ссылка; номер_строки; [номер_столбца]; [номер_области])	

Рис. 5.2. Начните ввод формулы на основе функции ИНДЕКС

Введите разделитель – точку с запятой, и переходите к набору аргумента номер_строки. Поскольку ваш массив одномерный (содержит по одному элементу в каждой строке), второй аргумент функции ИНДЕКС фактически соответствует номеру элемента в массиве. Вы должны выбрать объем продаж, одновременно отвечающий коду товара 2A35-2A36 и типу L. Функция ПОИСКПОЗ подходит для определения относительного положения элемента в списке. Вы можете поместить ПОИСКПОЗ в аргумент номер_строки функции ИНДЕКС (рис. 5.3). В аргументе

искомое_значение функции ПОИСКПОЗ вы сначала введете ссылку на ячейку с кодом товара, затем символ соединения – амперсанд &, и, наконец ссылку на ячейку типа товара (лево-/правосторонний). Обратите внимание на тип ссылок – они смешанные. Это опять же требуется для того, чтобы формула не «съехала» при протаскивании.¹

	A	B	C	D	E	F	G	H	I	J	K
1	Создание таблицы, которая ищет объем продаж на основе двух критериев										
2	Код товара	L/R?	Кол-во			Код товара	Кол-во				
3	2A45-2A46	R	18								
4	2A39-2A40	L	36		2A35-2A36	=ИНДЕКС(\$C\$3:\$C\$16;ПОИСКПОЗ(\$E4&F\$3					
5	2A35-2A36	L	30		2A38-2A39	30	36				ПОИСКПОЗ(искомое_значение; прос

Рис. 5.3. Создайте критерий поиска состоящий из двух элементов

Чтобы убедиться, что \$E4&F\$3 – одно значение, выделите этот элемент в формуле, и нажмите F9 (рис. 5.4). Используйте сочетание клавиш Ctrl+Z, чтобы отменить этот расчет и вернуться к набору формулы.

	Кол-во
Код товара	L R
2A35-2A36	=ИНДЕКС(\$C\$3:\$C\$16;ПОИСКПОЗ("2A35-2A36L"
2A38-2A39	30 36

Рис. 5.4. Используя F9, вы можете увидеть, что операция объединения создала единый аргумент *искомое_значение*

Введите точку с запятой и перейдите к вводу массива, в котором вы и будет искать относительную позицию элемента \$E4&F\$3. Поскольку у вас есть две колонки, вы также объедините их (рис. 5.5). В процессе соединения двух колонок вы собственно и выполнили операцию конкатенации массивов.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Создание таблицы, которая ищет объем продаж на основе двух критериев												
2	Код товара	L/R?	Кол-во			Код товара	Кол-во						
3	2A45-2A46	R	18										
4	2A39-2A40	L	36		2A35-2A36	=ИНДЕКС(\$C\$3:\$C\$16;ПОИСКПОЗ(\$E4&F\$3;\$A\$3:\$A\$16&B3:B16							
5	2A35-2A36	L	30		2A38-2A39	30	36						
6	2A38-2A39	R	36		2A39-2A40	36	35						
7	2A44-2A45	L	18		2A44-2A45	18	28						
8	2A48-2A49	L	20		2A45-2A46	24	18						
9	2A45-2A46	L	24		2A46-2A47	30	36						
10	2A48-2A49	R	24		2A48-2A49	20	24						
11	2A44-2A45	R	28										
12	2A46-2A47	L	30										
13	2A38-2A39	L	30										
14	2A35-2A36	R	35										
15	2A39-2A40	R	35										
16	2A46-2A47	R	36										
17													

Рис. 5.5. Аргумент *просматриваемый_массив* функции ПОИСПОЗ содержит оператор конкатенации массивов

Мы можете увидеть, что соединили две колонки и создали один массив, если выделите аргумент *просматриваемый_массив* и нажмете F9 (рис. 5.6). Неправда ли, это удивительно! Прямо в формуле, вы объединили два столбца в один. Не забудьте отменить расчет, нажав Ctrl+Z.

	Кол-во
Код товара	L R
2A35-2A36	=ИНДЕКС(\$C\$3:\$C\$16;ПОИСКПОЗ(\$E4&F\$3;{"2A45-2A46R";"2A39-2A40L";"2A35-2A36L";"2A38-2A39R";"2A44-2A45L";"2A48-2A49L";"2A45-2A46L";"2A46-2A47R";"2A38-2A39L";"2A39-2A40R";"2A46-2A47R";"2A35-2A36R";"2A39-2A40R";"2A46-2A47R"})
2A38-2A39	2A49L";"2A45-2A46L"; ПОИСКПОЗ(искомое_значение; <i>просматриваемый массив</i> ; [тип_сопоставления])
2A39-2A40	36 35

Рис. 5.6. Если вычислить аргумент *просматриваемый_массив*, вы увидите единый массив

Чтобы закончить ввод формулы ПОИСКПОЗ задайте аргумент *тип_соответствия* равным нулю (рис. 5.7); это укажет Excel искать точное совпадение. Введите две закрывающие скобки

¹ Подробнее см., например, [Относительные, абсолютные и смешанные ссылки на ячейки в Excel](#)

(напоминаю, что выбранный вами массив С3:С16 содержит один столбец, так что аргумент номер_столбца функции ИНДЕКС можно опустить).

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1	Создание таблицы, которая ищет объем продаж на основе двух критериев														
2	Код товара	L/R?	Кол-во			Код товара	L	R							
3	2A45-2A46	R	18			2A35-2A36	L								
4	2A39-2A40	L	36			2A38-2A39		30	36						
5	2A35-2A36	L	30			2A39-2A40		36	35						
6	2A38-2A39	R	36												
7	2A44-2A45	L	18												
8	2A48-2A49	L	20			2A45-2A46		24	18						

Рис. 5.7. Задайте тип_соответствия равным 0

Поскольку функция ПОИСКПОЗ изначально не запрограммирована на обработку операторов массива, нажмите Ctrl+Shift+Enter и скопируйте формулу на диапазон F4:G10 (рис. 5.8).

G7 : {=ИНДЕКС(\$C\$3:\$C\$16;ПОИСКПОЗ(\$E7&G\$3;\$A\$3:\$A\$16&\$B\$3:\$B\$16;0))}

	A	B	C	D	E	F	G	H	I	J
1	Создание таблицы, которая ищет объем продаж на основе двух критериев									
2	Код товара	L/R?	Кол-во			Код товара	L	R		
3	2A45-2A46	R	18			2A35-2A36	L	30	35	
4	2A39-2A40	L	36			2A38-2A39	L	30	36	
5	2A35-2A36	L	30			2A39-2A40	L	36	35	
6	2A38-2A39	R	36			2A44-2A45	L	18	28	
7	2A44-2A45	L	18			2A45-2A46	L	24	18	
8	2A48-2A49	L	20			2A46-2A47	L	30	36	
9	2A45-2A46	L	24			2A48-2A49	R	20	24	
10	2A48-2A49	R	24							
11	2A44-2A45	R	28							
12	2A46-2A47	L	30							
13	2A38-2A39	L	30							
14	2A35-2A36	R	35							
15	2A39-2A40	R	35							
16	2A46-2A47	R	36							

Рис. 5.8. Введите формул в ячейку F4 нажав Ctrl+Shift+Enter и скопируйте формулу по диапазону

Вы можете усовершенствовать вашу формулу конкатенации массивов. Представьте, например, вы соединили значения 30 и 20 и получили 3020. А если такое значение есть в одном из столбцов!? Чтобы застраховаться от подобных казусов введите какой-нибудь дополнительный редко встречающийся символ в качестве разделителя двух массивов (рис. 5.9).

Код товара	L/R?	Кол-во
2A35-2A36	L	30
2A38-2A39	R	36

Рис. 5.9. Дополнительный символ обезопасит формулу

А что, если у вас 3 критерия? Четыре? Вы можете соединить их все в формуле массива. Однако, операция конкатенации может потребоваться много времени для расчета формулы. Рассмотрите в качестве альтернативы функцию для работы с базой данных БИЗВЛЕЧЬ. БИЗВЛЕЧЬ является одной из самых привлекательных функций Excel. Если ваш набор данных содержит заголовки столбцов, БИЗВЛЕЧЬ можете сделать выборку на основе нескольких критериев. БД-функции могут работать как с И, так и с ИЛИ критериями. И критерии должны находиться в одной строке (рис. 5.10), ИЛИ критерии должны быть в разных строках (см. главу 11).

Как упоминалось ранее в этой книге, недостатком БД-функций являются проблемы с их копированием вдоль столбца, так как каждый раз нужно формировать область критериев отбора. Если вам не нужно копировать формулу в другие ячейки, функция БИЗВЛЕЧЬ – лучший выбор, так как ее проще написать и скорость ее работы выше. Если вам необходимо скопировать формулу в другие ячейки, вы можете использовать БИЗВЛЕЧЬ совместно с инструментом Таблица данных (см. описание к рис. 4.16 в главе 4).

Создание таблицы, которая ищет объем продаж на основе двух критериев						
Код товара	Кол-во	Qty				
Код товара	L/R?	Кол-во				
2A45-2A46	R	18	2A38-2A39	R	36	
2A39-2A40	L	36				
2A35-2A36	L	30				
2A38-2A39	R	36				
2A44-2A45	L	18				
2A48-2A49	L	20				
2A45-2A46	L	24				
2A48-2A49	R	24				
2A44-2A45	R	28				
2A46-2A47	L	30				
2A38-2A39	L	30				
2A35-2A36	R	35				
2A39-2A40	R	35				
2A46-2A47	R	36				

Рис. 5.10. Функция БИЗВЛЕЧЬ легче в создании и быстрее в работе, чем формула массива; однако, ее нельзя «протаскать» по столбцу

Еще одной альтернативой функции массива является вспомогательный столбец, в котором будут соединены критерии (так называемый объединенный ключ). Когда такой столбец создан, стандартная функция ВПР легко справится с извлечением нужных данных (рис. 5.11). Обратите внимание, вам достаточно вставить пустой столбец А и в ячейке А3 ввести формулу =В3&"|"&С3, которая легко копируется вниз по столбцу. Далее в аргументе *искомое_значение* функции ВПР создайте конкатенацию двух критериев \$F4&"|"&G\$3. Если вы можете позволить себе создание дополнительной колонки, решение с помощью ВПР – лучший выбор по сравнению с формулой массива. ВПР проще и работает быстрее.

Создание таблицы, которая ищет объем продаж на основе двух критериев						
Вспомогательная колонка	Код товара	L/R?	Кол-во	Код товара	Кол-во	
				Код товара	L	R
2A45-2A46 R	2A45-2A46	R	18	2A35-2A36	30	35
2A39-2A40 L	2A39-2A40	L	36	2A38-2A39	30	36
2A35-2A36 L	2A35-2A36	L	30	2A39-2A40	36	35
2A38-2A39 R	2A38-2A39	R	36	2A44-2A45	18	28
2A44-2A45 L	2A44-2A45	L	18	2A45-2A46	24	18
2A48-2A49 L	2A48-2A49	L	20	2A46-2A47	30	36
2A45-2A46 L	2A45-2A46	L	24	2A48-2A49	20	24
2A48-2A49 R	2A48-2A49	R	24			
2A44-2A45 R	2A44-2A45	R	28			
2A46-2A47 L	2A46-2A47	L	30			
2A38-2A39 L	2A38-2A39	L	30			
2A35-2A36 R	2A35-2A36	R	35			
2A39-2A40 R	2A39-2A40	R	35			
2A46-2A47 R	2A46-2A47	R	36			

Рис. 5.11. Если есть возможность, используйте вспомогательный столбец и функцию ВПР

Наконец, если вам не требуется немедленного обновления результатов, для создания перекрестного отчета вы можете использовать сводную таблицу (рис. 5.12). Обычно сводные таблицы используются для агрегирования данных, но они также могут использоваться и для особого их представления (упорядочения).

	A	B	C	D	E	F	G
1	Создание таблицы, которая ищет объем продаж на основе двух критериев						
2	Код товара	L/R?	Кол-во	Объем продаж			L/R?
3	2A45-2A46	R	18	Код товара	L	R	
4	2A39-2A40	L	36	2A35-2A36	30	35	
5	2A38-2A39	R	36	2A38-2A39	30	36	
6	2A44-2A45	L	18	2A39-2A40	36	35	
7	2A48-2A49	L	20	2A44-2A45	18	28	
8	2A45-2A46	L	24	2A45-2A46	24	18	
9	2A48-2A49	R	24	2A46-2A47	30	36	
10	2A44-2A45	R	28	2A48-2A49	20	24	
11	2A35-2A36	L	30				
12	2A46-2A47	L	30				
13	2A38-2A39	L	30				
14	2A35-2A36	R	35				
15	2A39-2A40	R	35				
16	2A46-2A47	R	36				
17							

Рис. 5.12. Если вы не нуждаетесь в немедленном обновлении результатов, возможно лучшее решение – сводная таблица

Рассмотрим еще один прием, позволяющий решить задачу выборки по двум критериям. Примените поиск на основании так называемого приблизительного совпадения; в этом случае значение аргумента *тип_сопоставления* функции ПОИСКПОЗ равен 1 или опущен (рис. 5.13).

ТИП СОПОСТАВЛЕНИЯ	ПОВЕДЕНИЕ
1 или опущен	Функция ПОИСКПОЗ находит наибольшее значение, которое меньше или равно значению аргумента искомое значение . Просматриваемый массив должен быть упорядочен по возрастанию: ..., -2, -1, 0, 1, 2, ..., A-Z, ЛОЖЬ, ИСТИНА.
0	Функция ПОИСКПОЗ находит первое значение, равное аргументу искомое значение . Просматриваемый массив может быть не упорядочен.
-1	Функция ПОИСКПОЗ находит наименьшее значение, которое больше или равно значению аргумента искомое значение . Просматриваемый массив должен быть упорядочен по убыванию: ИСТИНА, ЛОЖЬ, Z-A, ..., 2, 1, 0, -1, -2, ... и т. д.

Рис. 5.13. Аргумент *тип_сопоставления* функции ПОИСКПОЗ

Выборка на основании приблизительного совпадения работает гораздо быстрее, чем на основании точного соответствия. В первом случае Excel выполняет бинарный поиск, последовательно разбивая таблицу на половинки. Во втором случае Excel ищет, просматривая весь столбец сверху вниз. Отсортируйте по возрастанию сначала столбец *L/R?*, а затем столбец *Код товара*; таким образом, массив внутри формулы будет отсортирован по возрастанию. Далее, обратите внимание, что аргумент *тип_сопоставления* функции ПОИСКПОЗ опущен; по умолчанию поиск выполняется на основе приблизительного совпадения.

F4									
={ИНДЕКС(\$C\$3:\$C\$16;ПОИСКПОЗ(\$E4&F\$3;\$A\$3:\$A\$16&\$B\$3:\$B\$16))}									
A	B	C	D	E	F	G	H	I	J
1 Создание таблицы, которая ищет объем продаж на основе двух критериев									
2	Код товара	L/R?	Кол-во		Код товара	L	R	Qty	
3	2A35-2A36	L	30		2A35-2A36	L	R	30	35
4	2A35-2A36	R	35		2A38-2A39	L	R	30	36
5	2A38-2A39	L	30		2A39-2A40	L	R	36	35
6	2A38-2A39	R	36		2A44-2A45	L	R	18	28
7	2A39-2A40	L	36		2A45-2A46	L	R	24	18
8	2A39-2A40	R	35		2A46-2A47	L	R	30	36
9	2A44-2A45	L	18		2A48-2A49	L	R	20	24
10	2A44-2A45	R	28						
11	2A45-2A46	L	24						
12	2A45-2A46	R	18						
13	2A46-2A47	L	30						
14	2A46-2A47	R	36						
15	2A48-2A49	L	20						
16	2A48-2A49	R	24						
17									
18	1) Отсортировать по возрастанию колонку L/R?								
19	2) Отсортировать по возрастанию колонку Код товара								
20	3) Опустить аргумент тип_сопоставления функции ПОИСКПОЗ,								
21	что задаст поиск приблизительного соответствия								

Рис. 5.14. Если вы хотите ускорить работу формулы, отсортируйте два столбца и используйте поиск на основании приблизительного совпадения

Если у вас есть возможность отсортировать два столбца, вы можете упростить вашу формулу массива, используя вместо конструкции ИНДЕКС(ПОИСКПОЗ(...)) функцию ПРОСМОТР. Функция ПРОСМОТР может обрабатывать массив без нажатия Ctrl+Shift+Enter. Функция ПРОСМОТР ищет только приблизительное совпадение (поэтому-то исходные данные и должны быть отсортированы). Кроме того, эта функция имеет отдельные аргументы для столбца в котором вы ищите совпадения, названном *просматриваемый_вектор*, и для столбца, который содержит значения, которые вы хотите получить – *вектор_результатов* (рис. 5.15).

Код товара	L	R
2A35-2A36	=ПРОСМОТР(\$E4&F\$3;\$A\$3:\$A\$16&\$B\$3:\$B\$16;\$C\$3:\$C\$16)	
2A38-2A39	ПРОСМОТР(искмое_значение; просматриваемый_вектор; [вектор_результатов])	

Рис. 5.15. Функция ПРОСМОТР ищет приблизительное совпадение и может обрабатывать массив без Ctrl+Shift+Enter

Предварительная сортировка ускорит и работу функции ВПР, описанной на рис. 5.11. Если вы отсортируете по вспомогательному столбцу, вы можете использовать приблизительное совпадение, опустив четвертый аргумент функции ВПР: =ВПР(\$F4&"|"&G\$3;\$A\$3:\$D\$16;4) (рис. 5.16). Для больших наборов данных, переход от поиска точного соответствия к приблизительному может значительно сократить время расчета формулы.

G4							
=ВПР(\$F4&" "&G\$3;\$A\$3:\$D\$16;4)							
A	B	C	D	E	F	G	H
1 Создание таблицы, которая ищет объем продаж на основе двух критериев							
2	Вспомогательная колонка	Код товара	L/R?	Кол-во		Кол-во	
3	2A35-2A36 L	2A35-2A36	L	30	Код товара	L	R
4	2A35-2A36 R	2A35-2A36	R	35	2A35-2A36	30	35
5	2A38-2A39 L	2A38-2A39	L	30	2A38-2A39	30	36
6	2A38-2A39 R	2A38-2A39	R	36	2A39-2A40	36	35
7	2A39-2A40 L	2A39-2A40	L	36	2A44-2A45	18	28
8	2A39-2A40 R	2A39-2A40	R	35	2A45-2A46	24	18
9	2A44-2A45 L	2A44-2A45	L	18	2A46-2A47	30	36
10	2A44-2A45 R	2A44-2A45	R	28	2A48-2A49	20	24
11	2A45-2A46 L	2A45-2A46	L	24			
12	2A45-2A46 R	2A45-2A46	R	18			
13	2A46-2A47 L	2A46-2A47	L	30			
14	2A46-2A47 R	2A46-2A47	R	36			
15	2A48-2A49 L	2A48-2A49	L	20			
16	2A48-2A49 R	2A48-2A49	R	24			
17							

Рис. 5.16. Ускоряем функцию ВПР; отличие от рис. 5.11 в том, что диапазон A3:D16 отсортирован по столбцу A, а в функции ВПР опущен четвертый аргумент

Ранее в этой книге, вы познакомились с тремя функциями, которые могут обрабатывать массивы без Ctrl+Shift+Enter: СУММПРОИЗВ, АГРЕГАТ и ПРОСМОТР. Четвертой такой функцией является ИНДЕКС.

Следующий пример выглядит немного искусственно, но он показывает, как оператор массива, помещенный в аргумент функции ИНДЕКС может обрабатываться без Ctrl+Shift+Enter. Вернитесь к рис. 5.7, где приведена формула массива для поиска по двум критериям. Поскольку операция с массивами была расположена в аргументе *просматриваемый_массив* функции ПОИСКПОЗ, вам пришлось использовать Ctrl+Shift+Enter (функция ПОИСКПОЗ не была изначально запрограммирована для операций с массивами). Повторное использование функции ИНДЕКС позволяет вам изменить формулу, которая теперь не потребует нажатия Ctrl+Shift+Enter.

Обычно функция ИНДЕКС ищет один элемент в двумерном (строку и столбец) или одномерном массиве (только строку или только столбец). Но ИНДЕКС может также искать целиком строку или столбец. В этом случае, функция вернет массив элементов. Синтаксис функции

ИНДЕКС(массив;номер_строки;[номер_столбца])

Для поиска всего столбца разместите в аргументе *номер_строки* ноль или оставьте его пустым. Это укажет функции ИНДЕКС получить весь столбец, т.е. все строки. Это означает, что вы можете разместить оператор конкатенации массивов в аргументе *массив* функции ИНДЕКС и оставить аргумент *номер_строки* пустым (рис. 5.17). Функция ИНДЕКС вернет массив, состоящий из всех объединенных элементов и предоставит этот массив в распоряжение функции ПОИСКПОЗ. Но... теперь ввод формулы не потребует нажатия Ctrl+Shift+Enter.

Код товара	L	R
2A35-2A36		
2A38-2A39	30	36

ИНДЕКС(массив; номер_строки; [номер_столбца])

Рис. 5.17. Разместите оператор конкатенации массивов в аргументе *массив* функции ИНДЕКС и оставьте аргумент *номер_строки* пустым

Следующие пять аргументов четырех функций могут обрабатывать массивы без Ctrl+Shift+Enter:

- *Массив1*, *массив2* и др. аргументы функции СУММПРОИЗВ
- *Просматриваемый_вектор* функции ПРОСМОТР
- *Вектор_результатов* функции ПРОСМОТР
- *Массив* функции ИНДЕКС
- *Массив* функции АГРЕГАТ для функций от 14 до 19